

Apèndix

Àlgebra lineal amb wxMaxima

Objectius

1. Definir matrius amb wxMaxima.
2. Aplicar amb wxMaxima operacions amb matrius.
3. Aplicar transformacions elementals de matrius.
4. Calcular el determinant d'una matriu quadrada.
5. Interpretar les propietats del determinant d'una matriu quadrada.
6. Plantejar i resoldre sistemes d'equacions lineals amb wxMaxima.
7. Discutir i resoldre sistemes d'equacions lineals.

Continguts

- A-1. Matrius. Operacions amb matrius.
- A-2 Transformacions elementals de matrius.
- A-3 Rang d'una matriu. Inversa d'una matriu quadrada regular. Factorització LU.
- A-4. Determinant d'una matriu quadrada. Propietats. Càlcul
- A-3. Sistemes d'equacions lineals.

Referències

- AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.
- G07 GLASNER, MOSES (2007)
A Maxima Guide for Calculus Students
- RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J. RAFAEL (2005)
Introducción a Maxima
- RG07 RODRÍGUEZ GALVÁN, J. RAFAEL (2007)
Maxima con wxMaxima: software libre en el aula de matemáticas
- RR08 RODRÍGUEZ RIOTORTO, MARIO (2008)
Primeros pasos en Maxima

- RV09 RODRÍGUEZ, MARIO; VILLATE, JAIME (2009)
Manual de Maxima.
- VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

A-1.- Matrius. Operacions amb matrius

Els continguts d'aquest apartat es desenvolupen al fitxer **Appendix-1.wxm**.

En aquest apartat es veurà la definició de matrius amb wxMaxima i les diferents operacions amb matrius.

A.1.1. Definició de matrius i submatrius.

Definició de matrius:

```
(%i1) 'A:=matrix( [a11, a12, a13, a14],  
                  [a21, a22, a23, a24],  
                  [a31, a32, a33, a34]  
                  );  
(%o1) A = 
$$\begin{bmatrix} a_{11} & a_{12} & a_{13} & a_{14} \\ a_{21} & a_{22} & a_{23} & a_{24} \\ a_{31} & a_{32} & a_{33} & a_{34} \end{bmatrix}$$

```

Explicitar una fila o una columna d'una matriu:

```
(%i2) a1=row(A,1);    a3=row(A,3);  
(%o2) a1 = 
$$\begin{bmatrix} a_{11} & a_{12} & a_{13} & a_{14} \end{bmatrix}$$
  
(%o3) a3 = 
$$\begin{bmatrix} a_{31} & a_{32} & a_{33} & a_{34} \end{bmatrix}$$
  
  
(%i4) a[1]=col(A,1);    a[4]=col(A,4);  
(%o4) a1 = 
$$\begin{bmatrix} a_{11} \\ a_{21} \\ a_{31} \end{bmatrix}$$
  
(%o5) a4 = 
$$\begin{bmatrix} a_{14} \\ a_{24} \\ a_{34} \end{bmatrix}$$

```

Obtenció de submatrius per eliminació d'una fila:

```
(%i6) 'A1=submatrix(1,A);  
(%o6) A1 = 
$$\begin{bmatrix} a_{21} & a_{22} & a_{23} & a_{24} \\ a_{31} & a_{32} & a_{33} & a_{34} \end{bmatrix}$$

```

Obtenció de submatrius per eliminació d'una columna:

```
(%i7) 'A2=submatrix(A,2);
```

$$(\%o7) A2 = \begin{bmatrix} a_{11} & a_{13} & a_{14} \\ a_{21} & a_{23} & a_{24} \\ a_{31} & a_{33} & a_{34} \end{bmatrix}$$

Obtenció de submatrius per eliminació d'una fila i una columna:

```
(%i8) 'A12=submatrix(1,A,2);
```

$$(\%o8) A12 = \begin{bmatrix} a_{21} & a_{23} & a_{24} \\ a_{31} & a_{33} & a_{34} \end{bmatrix}$$

Obtenció de submatrius per eliminació de més d'una fila i una columna:

```
(%i9) 'A123=submatrix(1, 2, A, 3);
```

$$(\%o9) A123 = \begin{bmatrix} a_{31} & a_{32} & a_{34} \end{bmatrix}$$

Definició d'un vector columna (observeu que prèviament s'ha de carregar el paquet "eigen"):

```
(%i10) load ("eigen")$
      X3=X3:columnvector([x1, x2, x3]);
```

$$(\%o11) X3 = \begin{bmatrix} x1 \\ x2 \\ x3 \end{bmatrix}$$

Definició d'una matriu per una funció dels índex de fila i de columna:

```
(%i12) f[i,j]:=2*i+3*j;
      B1=B1:genmatrix(f,4,3);
```

$$(\%o12) f_{i,j} := 2i + 3j$$

$$(\%o13) B1 = \begin{bmatrix} 5 & 8 & 11 \\ 7 & 10 & 13 \\ 9 & 12 & 15 \\ 11 & 14 & 17 \end{bmatrix}$$

```
(%i14) g[i,j]:=1/(2*i+3*j-1);
      B2=B2:genmatrix(g,3,4);
```

```
(%o14)  $g_{i,j} := \frac{1}{2i+3j-1}$ 
```

```
(%o15) B2 = 
$$\begin{bmatrix} \frac{1}{4} & \frac{1}{7} & \frac{1}{10} & \frac{1}{13} \\ \frac{1}{6} & \frac{1}{9} & \frac{1}{12} & \frac{1}{15} \\ \frac{1}{8} & \frac{1}{11} & \frac{1}{14} & \frac{1}{17} \end{bmatrix}$$

```

Definició d'una matriu diagonal:

```
(%i16) 'D5=D5:diag_matrix(a11,a22,a33,a44,a55);
```

```
(%o16) D5 = 
$$\begin{bmatrix} a11 & 0 & 0 & 0 & 0 \\ 0 & a22 & 0 & 0 & 0 \\ 0 & 0 & a33 & 0 & 0 \\ 0 & 0 & 0 & a44 & 0 \\ 0 & 0 & 0 & 0 & a55 \end{bmatrix}$$

```

Definició de les matrius unitat (o identitat):

```
(%i17) I[2]=I2:ident(2);      I[4]=I4:ident(4);
```

```
(%o17) I2 = 
$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

```

```
(%o18) I4 = 
$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```

Definició d'una matriu nul·la d'un cert ordre:

```
(%i19) O34=O34:zeromatrix(3,4);
```

```
(%o19) O34 = 
$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

```

Càlcul de la traça d'una matriu quadrada:

```
(%i20) 'C:=C:matrix( [a11, a12], [a21, a22] );
      load("nchrpl")$      'tr(C)=mattrace(C);
(%o20) C = 
$$\begin{bmatrix} a_{11} & a_{12} \\ a_{21} & a_{22} \end{bmatrix}$$

(%o22) tr(C) = a22 + a11
```

A.1.2 Operacions algebraiques amb matrius.

Suma de matrius de la mateixa estructura:

```
(%i1) 'A1=A1:matrix( [a11, a12, a13, a14], [a21, a22, a23, a24], [a31, a32, a33, a34] )$
      'A2=A2:matrix( [b11, b12, b13, b14], [b21, b22, b23, b24], [b31, b32, b33, b34] )$
(%i3) '(A1+A2)=A1+A2;
(%o3) A2 + A1 = 
$$\begin{bmatrix} b_{11}+a_{11} & b_{12}+a_{12} & b_{13}+a_{13} & b_{14}+a_{14} \\ b_{21}+a_{21} & b_{22}+a_{22} & b_{23}+a_{23} & b_{24}+a_{24} \\ b_{31}+a_{31} & b_{32}+a_{32} & b_{33}+a_{33} & b_{34}+a_{34} \end{bmatrix}$$

```

Producte d'una matriu per un escalar:

```
(%i4) '(%lambda*A1)=%lambda*A1;
(%o4)  $\lambda$  A1 = 
$$\begin{bmatrix} \lambda a_{11} & \lambda a_{12} & \lambda a_{13} & \lambda a_{14} \\ \lambda a_{21} & \lambda a_{22} & \lambda a_{23} & \lambda a_{24} \\ \lambda a_{31} & \lambda a_{32} & \lambda a_{33} & \lambda a_{34} \end{bmatrix}$$

```

Combinació lineal de matrius:

```
(%i5) '(%lambda*A1 + %mu*A2)=%lambda*A1 + %mu*A2;
(%o5)  $\mu$  A2 +  $\lambda$  A1 = 
$$\begin{bmatrix} \mu b_{11}+\lambda a_{11} & \mu b_{12}+\lambda a_{12} & \mu b_{13}+\lambda a_{13} & \mu b_{14}+\lambda a_{14} \\ \mu b_{21}+\lambda a_{21} & \mu b_{22}+\lambda a_{22} & \mu b_{23}+\lambda a_{23} & \mu b_{24}+\lambda a_{24} \\ \mu b_{31}+\lambda a_{31} & \mu b_{32}+\lambda a_{32} & \mu b_{33}+\lambda a_{33} & \mu b_{34}+\lambda a_{34} \end{bmatrix}$$

```

Per al producte de matrius cal fer observar que en alguns programes el símbol és l'asterisc (*). Vegem què passa amb wxMaxima:

```
(%i6) '(A1*A2)=A1*A2;
(%o6) A1 A2 = 
$$\begin{bmatrix} a_{11} b_{11} & a_{12} b_{12} & a_{13} b_{13} & a_{14} b_{14} \\ a_{21} b_{21} & a_{22} b_{22} & a_{23} b_{23} & a_{24} b_{24} \\ a_{31} b_{31} & a_{32} b_{32} & a_{33} b_{33} & a_{34} b_{34} \end{bmatrix}$$

```

Òbviament aquesta operació no és el producte de matrius, per a les quals cal que el nombre de columnes de la primera sigui igual al nombre de files de la segona. Vegem la sintaxi amb wxMaxima:

```
(%i7) 'A3:=A3:matrix([c11, c12, c13], [c21, c22, c23], [c31, c32, c33], [c41, c42, c43] );
```

$$(\%o7) A3 = \begin{bmatrix} c11 & c12 & c13 \\ c21 & c22 & c23 \\ c31 & c32 & c33 \\ c41 & c42 & c43 \end{bmatrix}$$

```
(%i8) '(A1.A3)=A1.A3;
```

```
(%o8) A1 . A3 =
```

$$\begin{bmatrix} a14 c41 + a13 c31 + a12 c21 + a11 c11 & a14 c42 + a13 c32 + a12 c22 + a11 c12 & a14 c43 + a13 c33 + a12 c23 + a11 c13 \\ a24 c41 + a23 c31 + a22 c21 + a21 c11 & a24 c42 + a23 c32 + a22 c22 + a21 c12 & a24 c43 + a23 c33 + a22 c23 + a21 c13 \\ a34 c41 + a33 c31 + a32 c21 + a31 c11 & a34 c42 + a33 c32 + a32 c22 + a31 c12 & a34 c43 + a33 c33 + a32 c23 + a31 c13 \end{bmatrix}$$

```
(%i9) '(A3.A1)=A3.A1;
```

```
(%o9) A3 . A1 =
```

$$\begin{bmatrix} a31 c13 + a21 c12 + a11 c11 & a32 c13 + a22 c12 + a12 c11 & a33 c13 + a23 c12 + a13 c11 & a34 c13 + a24 c12 + a14 c11 \\ a31 c23 + a21 c22 + a11 c21 & a32 c23 + a22 c22 + a12 c21 & a33 c23 + a23 c22 + a13 c21 & a34 c23 + a24 c22 + a14 c21 \\ a31 c33 + a21 c32 + a11 c31 & a32 c33 + a22 c32 + a12 c31 & a33 c33 + a23 c32 + a13 c31 & a34 c33 + a24 c32 + a14 c31 \\ a31 c43 + a21 c42 + a11 c41 & a32 c43 + a22 c42 + a12 c41 & a33 c43 + a23 c42 + a13 c41 & a34 c43 + a24 c42 + a14 c41 \end{bmatrix}$$

Serveixi això per recordar que el producte de matrius no és commutatiu.

Producte d'una matriu per un vector columna:

```
(%i10) load ("eigen")$
```

```
X4=X4:columnvector([x1, x2, x3, x4])$
```

```
A1X4=A1X4:A1.X4;
```

$$(\%o12) A1X4 = \begin{bmatrix} a14 x4 + a13 x3 + a12 x2 + a11 x1 \\ a24 x4 + a23 x3 + a22 x2 + a21 x1 \\ a34 x4 + a33 x3 + a32 x2 + a31 x1 \end{bmatrix}$$

Transposició de matrius:

```
(%i13) 'A1T=A1T:transpose(A1);
```

$$(\%o13) A1T = \begin{bmatrix} a11 & a21 & a31 \\ a12 & a22 & a32 \\ a13 & a23 & a33 \\ a14 & a24 & a34 \end{bmatrix}$$

A.1.3 Concatenació de matrius.

A partir d'unes matrius inicials es poden construir noves matrius per concatenació de files, és a dir, escriure a continuació de cada fila d'una matriu les files de la segona:

```
(%i14) 'A12=addcol(A1, A2);
```

$$(\%o14) \ A12 = \begin{bmatrix} a_{11} & a_{12} & a_{13} & a_{14} & b_{11} & b_{12} & b_{13} & b_{14} \\ a_{21} & a_{22} & a_{23} & a_{24} & b_{21} & b_{22} & b_{23} & b_{24} \\ a_{31} & a_{32} & a_{33} & a_{34} & b_{31} & b_{32} & b_{33} & b_{34} \end{bmatrix}$$

També es poden construir noves matrius per concatenació de columnes, és a dir, escriure a continuació de cada columna d'una matriu les columnes de la segona:

```
(%i15) 'A22=addrow(A1, A2);
```

$$(\%o15) \ A22 = \begin{bmatrix} a_{11} & a_{12} & a_{13} & a_{14} \\ a_{21} & a_{22} & a_{23} & a_{24} \\ a_{31} & a_{32} & a_{33} & a_{34} \\ b_{11} & b_{12} & b_{13} & b_{14} \\ b_{21} & b_{22} & b_{23} & b_{24} \\ b_{31} & b_{32} & b_{33} & b_{34} \end{bmatrix}$$

A.1.4 Matrius per blocs.

A partir de matrius quadrades es poden construir noves matrius amb l'estructura anomenada “diagonal per blocs” o “bloc-diagonal”:

```
(%i16) load("diag")$
```

```
A3=A3:matrix([a11, a12, a13], [a21, a22, a23], [a31, a32, a33])$
```

```
A4=A4:matrix([b11, b12], [b21, b22])$
```

```
(%i19) diag([A3,A4]);
```

$$(\%o19) \ \begin{bmatrix} a_{11} & a_{12} & a_{13} & 0 & 0 \\ a_{21} & a_{22} & a_{23} & 0 & 0 \\ a_{31} & a_{32} & a_{33} & 0 & 0 \\ 0 & 0 & 0 & b_{11} & b_{12} \\ 0 & 0 & 0 & b_{21} & b_{22} \end{bmatrix}$$

(%i20) diag([A4,A3]);

(%o20)
$$\begin{bmatrix} b_{11} & b_{12} & 0 & 0 & 0 \\ b_{21} & b_{22} & 0 & 0 & 0 \\ 0 & 0 & a_{11} & a_{12} & a_{13} \\ 0 & 0 & a_{21} & a_{22} & a_{23} \\ 0 & 0 & a_{31} & a_{32} & a_{33} \end{bmatrix}$$

A-2.- Transformacions elementals de matrius

Els continguts d'aquest apartat es desenvolupen al fitxer **Appendix-2.wxm**.

A.2.1. Transformacions elementals de files i columnes

S'anomenen transformacions elementals de matrius les següents:

1. Intercanvi de dues files o dues columnes.
2. Multiplicació d'una fila o columna per un escalar no nul.
3. Sumar a una fila o columna un múltiple escalar (no nul) d'una altra fila o columna.

Vegem la sintaxi amb wxMaxima.

1. Intercanvi de dues files o dues columnes:

```
(%i1) 'A:=A:matrix( [a11, a12, a13, a14],  
                    [a21, a22, a23, a24],  
                    [a31, a32, a33, a34] );
```

```
(%o1) A = 
$$\begin{bmatrix} a_{11} & a_{12} & a_{13} & a_{14} \\ a_{21} & a_{22} & a_{23} & a_{24} \\ a_{31} & a_{32} & a_{33} & a_{34} \end{bmatrix}$$

```

```
(%i2) A1:=A1:rowswap(A, 1, 3);
```

```
(%o2) A1 = 
$$\begin{bmatrix} a_{31} & a_{32} & a_{33} & a_{34} \\ a_{21} & a_{22} & a_{23} & a_{24} \\ a_{11} & a_{12} & a_{13} & a_{14} \end{bmatrix}$$

```

Val a dir que aquesta transformació s'obté premultiplicant la matriu per la matriu unitat del mateix nombre de files en la qual s'hagi aplicat la transformació que es vol aplicar en la matriu. En efecte:

```
(%i3) I3a:=I3a:rowswap(ident(3), 1, 3);
```

```
(%o3) I3a = 
$$\begin{bmatrix} 0 & 0 & 1 \\ 0 & 1 & 0 \\ 1 & 0 & 0 \end{bmatrix}$$

```

```
(%i4) I3a.A;
```

```
(%o4) 
$$\begin{bmatrix} a_{31} & a_{32} & a_{33} & a_{34} \\ a_{21} & a_{22} & a_{23} & a_{24} \\ a_{11} & a_{12} & a_{13} & a_{14} \end{bmatrix}$$

```

Intercanvi de columnes:

```
(%i5) A2=A2:columnswap(A, 1, 3);
```

$$(\%o5) A2 = \begin{bmatrix} a_{13} & a_{12} & a_{11} & a_{14} \\ a_{23} & a_{22} & a_{21} & a_{24} \\ a_{33} & a_{32} & a_{31} & a_{34} \end{bmatrix}$$

Aquesta transformació s'obté postmultiplicant la matriu per la matriu unitat del mateix nombre de columnes en la qual s'hagi aplicat la transformació que es vol aplicar en la matriu. En efecte:

```
(%i6) I4a=I4a:columnswap(ident(4), 1, 3);
```

$$(\%o6) I4a = \begin{bmatrix} 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
(%i7) A.I4a;
```

$$(\%o7) \begin{bmatrix} a_{13} & a_{12} & a_{11} & a_{14} \\ a_{23} & a_{22} & a_{21} & a_{24} \\ a_{33} & a_{32} & a_{31} & a_{34} \end{bmatrix}$$

2. Multiplicació d'una fila o columna per un escalar no nul.

No hi ha una instrucció de wxMaxima que realitzi aquesta transformació. Per això s'ha de recórrer a la metodologia de premultiplicació (files) o postmultiplicació (columnes) comentada al cas anterior:

```
(%i8) I3b=I3b:matrix( [1,0,0], [0, %lambda,0], [0,0,1] );
```

$$(\%o8) I3b = \begin{bmatrix} 1 & 0 & 0 \\ 0 & \lambda & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

(%i9) I3b.A;

(%o9)
$$\begin{bmatrix} a_{11} & a_{12} & a_{13} & a_{14} \\ \lambda a_{21} & \lambda a_{22} & \lambda a_{23} & \lambda a_{24} \\ a_{31} & a_{32} & a_{33} & a_{34} \end{bmatrix}$$

(%i10) I4b:=I4b:matrix([1,0,0,0], [0,%lambda,0,0], [0,0,1,0], [0,0,0,1]);

(%o10) I4b =
$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & \lambda & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

(%i11) A.I4b;

(%o11)
$$\begin{bmatrix} a_{11} & \lambda a_{12} & a_{13} & a_{14} \\ a_{21} & \lambda a_{22} & a_{23} & a_{24} \\ a_{31} & \lambda a_{32} & a_{33} & a_{34} \end{bmatrix}$$

3. Sumar a una fila o columna un múltiple escalar (no nul) d'una altra fila o columna.

(%i12) rowop(A,1,2,-%lambda);

(%o12)
$$\begin{bmatrix} \lambda a_{21} + a_{11} & \lambda a_{22} + a_{12} & \lambda a_{23} + a_{13} & \lambda a_{24} + a_{14} \\ a_{21} & a_{22} & a_{23} & a_{24} \\ a_{31} & a_{32} & a_{33} & a_{34} \end{bmatrix}$$

Vegem també la seva implementació amb la premultiplicació per la transformació feta en la matriu unitat:

(%i13) I3c:=I3c:rowop(ident(3), 1, 2, -%lambda);

(%o13) I3c =
$$\begin{bmatrix} 1 & \lambda & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

(%i14) I3c.A;

(%o14)
$$\begin{bmatrix} \lambda a_{21} + a_{11} & \lambda a_{22} + a_{12} & \lambda a_{23} + a_{13} & \lambda a_{24} + a_{14} \\ a_{21} & a_{22} & a_{23} & a_{24} \\ a_{31} & a_{32} & a_{33} & a_{34} \end{bmatrix}$$

I ara amb una columna:

```
(%i15) columnop(A,1,2,-%lambda);
```

$$(\%o15) \begin{bmatrix} \lambda a_{12}+a_{11} & a_{12} & a_{13} & a_{14} \\ \lambda a_{22}+a_{21} & a_{22} & a_{23} & a_{24} \\ \lambda a_{32}+a_{31} & a_{32} & a_{33} & a_{34} \end{bmatrix}$$

```
(%i16) I4c=I4c:columnop(ident(4), 1, 2, -%lambda);
```

$$(\%o16) I4c = \begin{bmatrix} 1 & 0 & 0 & 0 \\ \lambda & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
(%i17) A.I4c;
```

$$(\%o17) \begin{bmatrix} \lambda a_{12}+a_{11} & a_{12} & a_{13} & a_{14} \\ \lambda a_{22}+a_{21} & a_{22} & a_{23} & a_{24} \\ \lambda a_{32}+a_{31} & a_{32} & a_{33} & a_{34} \end{bmatrix}$$

A.2.2. Obtenció de matrius esglaonades i matrius esglaonades reduïdes.

Aquestes matrius s'obtenen amb les instruccions “triangularize”, resultant una matriu triangular superior, i “echelon” que dona una matriu triangular superior reduïda. Vegem la sintaxi mitjançant uns exemples.

```
(%i1) 'A=A:matrix( [3, 4, -2, 6],
                    [5, 2, 0, -1],
                    [7, 2, 8, 9] );
```

$$(\%o1) A = \begin{bmatrix} 3 & 4 & -2 & 6 \\ 5 & 2 & 0 & -1 \\ 7 & 2 & 8 & 9 \end{bmatrix}$$

```
(%i2) triangularize(A);
```

$$(\%o2) \begin{bmatrix} 5 & 2 & 0 & -1 \\ 0 & 14 & -10 & 33 \\ 0 & 0 & 104 & 172 \end{bmatrix}$$

(%i3) echelon(A);

(%o3)
$$\begin{bmatrix} 1 & \frac{2}{5} & 0 & -\frac{1}{5} \\ 0 & 1 & -\frac{5}{7} & \frac{33}{14} \\ 0 & 0 & 1 & \frac{43}{26} \end{bmatrix}$$

(%i4) fpprintprec:5\$
echelon(A), numer;

(%o5)
$$\begin{bmatrix} 1 & 0.4 & 0 & -0.2 \\ 0 & 1 & -0.714 & 2.3571 \\ 0 & 0 & 1 & 1.6538 \end{bmatrix}$$

A-3.- Rang d'una matriu. Inversa d'una matriu. Factorització LU.

Els continguts d'aquest apartat es desenvolupen al fitxer **Apendix-3.wxm**.

A.3.1. Rang d'una matriu

Per calcular el rang d'una matriu cal aplicar la instrucció “rank”:

```
(%i1) 'A=A:matrix( [3, 4, -2, 6],  
                  [5, 2, 0, -1],  
                  [7, 2, 8, 9] );  
(%o1) A =  $\begin{bmatrix} 3 & 4 & -2 & 6 \\ 5 & 2 & 0 & -1 \\ 7 & 2 & 8 & 9 \end{bmatrix}$ 
```

```
(%i2) r=rank(A);  
(%o2) r = 3
```

A.3.2. Inversa d'una matriu

Per calcular la inversa d'una matriu quadrada regular cal verificar primer que el seu rang és l'adequat i aplicar després la instrucció “inv”:

```
(%i1) 'A=A:matrix([2,5],[1,3]);  
(%o1) A =  $\begin{bmatrix} 2 & 5 \\ 1 & 3 \end{bmatrix}$ 
```

```
(%i2) rank(A);  
(%o2) 2
```

```
(%i3) '(inv(A))=invA:invert(A);  
(%o3) inv(A) =  $\begin{bmatrix} 3 & -5 \\ -1 & 2 \end{bmatrix}$ 
```

Com es pot veure a continuació el càlcul mitjançant els mètodes “manuels” tradicionals pot resultar complicat:

```
(%i4) fpprintprec:5$
      'C=C:matrix( [3.123, 4.543, -2.541, 4.543],
                  [5.027, 2.977, 0.891, -3.982],
                  [7.002, 2.804, 8.492, -2.304],
                  [2.804, 8.492, -2.304, 7.004] );
```

```
(%o5) C = 
$$\begin{bmatrix} 3.123 & 4.543 & -2.541 & 4.543 \\ 5.027 & 2.977 & 0.891 & -3.982 \\ 7.002 & 2.804 & 8.492 & -2.304 \\ 2.804 & 8.492 & -2.304 & 7.004 \end{bmatrix}$$

```

```
(%i6) invC=invert(C);
```

```
(%o6) invC = 
$$\begin{bmatrix} 0.391 & 0.01 & 0.0535 & -0.23 \\ -0.34 & 0.115 & -0.0398 & 0.273 \\ -0.155 & -0.0932 & 0.103 & 0.0813 \\ 0.205 & -0.174 & 0.0609 & -0.0688 \end{bmatrix}$$

```

A.3.3. Factorització LU

Per calcular la descomposició LU d'una matriu quadrada regular, és a dir, dues matrius la primera de les quals (L) és triangular inferior i la segona de les quals (U) és triangular superior, que multiplicades en aquest ordre donen la matriu A, hi ha implementada en wxMaxima una instrucció que dóna el resultat corresponent. Vegem un exemple.

```
(%i1) 'A=A:matrix([1,2,-3],[-3,-4,13],[2,1,-5]);
```

```
(%o1) A = 
$$\begin{bmatrix} 1 & 2 & -3 \\ -3 & -4 & 13 \\ 2 & 1 & -5 \end{bmatrix}$$

```

Vegem que la matriu és regular:

```
(%i2) r=rank(A);
```

```
(%o2) r = 3
```

I ara apliquem la instrucció que calcula aquests factors:

```
(%i3) lu_factor(A, generalring)$
      get_lu_factors (%);
```

$$(\%o4) \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ -3 & 1 & 0 \\ 2 & -\frac{3}{2} & 1 \end{bmatrix}, \begin{bmatrix} 1 & 2 & -3 \\ 0 & 2 & 4 \\ 0 & 0 & 7 \end{bmatrix}$$

La segona matriu és la triangular inferior (L) i la tercera és la triangular superior (U). En efecte:

```
(%i5) L=L:matrix([1,0,0],[-3,1,0],[2,-3/2,1])$
      U=U:matrix([1,2,-3],[0,2,4],[0,0,7])$
      L.U;
```

$$(\%o7) \begin{bmatrix} 1 & 2 & -3 \\ -3 & -4 & 13 \\ 2 & 1 & -5 \end{bmatrix}$$

A-4.- Determinant d'una matriu quadrada.

Els continguts d'aquest apartat es desenvolupen al fitxer **Apendix-4.wxm**.

Vegem el desenvolupament del càlcul del determinant d'una matriu quadrada d'ordre n , en els casos $n = 2$ i $n = 3$:

```
(%i1) 'A2=A2:matrix([a11,a12],[a21,a22]);    detA2=determinant(A2);  
      det(A2)=determinant(A2);  
  
(%o1) A2=
$$\begin{bmatrix} a_{11} & a_{12} \\ a_{21} & a_{22} \end{bmatrix}$$
  
  
(%o2) detA2 = a11 a22 - a12 a21  
  
(%o3) det(
$$\begin{bmatrix} a_{11} & a_{12} \\ a_{21} & a_{22} \end{bmatrix}$$
) = a11 a22 - a12 a21  
  
(%i4) 'A3=A3:matrix([a11,a12,a13],[a21,a22,a23],[a31,a32,a33]);    detA3=determinant(A3);    expand(%);  
(%o4) A3=
$$\begin{bmatrix} a_{11} & a_{12} & a_{13} \\ a_{21} & a_{22} & a_{23} \\ a_{31} & a_{32} & a_{33} \end{bmatrix}$$
  
  
(%o5) detA3 = a11 (a22 a33 - a23 a32) - a12 (a21 a33 - a23 a31) + a13 (a21 a32 - a22 a31)  
(%o6) detA3 = a11 a22 a33 - a12 a21 a33 - a11 a23 a32 + a13 a21 a32 + a12 a23 a31 - a13 a22 a31
```

Com és sabut, el determinant d'una matriu triangular és igual al producte dels elements de la diagonal. En efecte:

```
(%i7) 'B3=B3:matrix([a11,a12,a13],[0,a22,a23],[0,0,a33]);  
      detB3=determinant(B3);  
  
(%o7) B3=
$$\begin{bmatrix} a_{11} & a_{12} & a_{13} \\ 0 & a_{22} & a_{23} \\ 0 & 0 & a_{33} \end{bmatrix}$$
  
  
(%o8) detB3 = a11 a22 a33  
  
(%i9) 'B4=B4:matrix([a11,a12,a13,a14],[0,a22,a23,a24],[0,0,a33,a34],[0,0,0,a44]);  
      detB4=determinant(B4);  
  
(%o9) B4=
$$\begin{bmatrix} a_{11} & a_{12} & a_{13} & a_{14} \\ 0 & a_{22} & a_{23} & a_{24} \\ 0 & 0 & a_{33} & a_{34} \\ 0 & 0 & 0 & a_{44} \end{bmatrix}$$
  
  
(%o10) detB4 = a11 a22 a33 a44
```

Vegem ara un exemple numèric:

```
(%i1) 'C1=C1: matrix( [1.01,2.02,0.99,0.01,0.01],
                      [1.02,1.01,-1.01,2.02,0.01],
                      [1.99,2.99,0.01,1.01,0.02],
                      [-1.01,0.02,1.02,-2.01,0.03],
                      [0.24, -0.972, 1.23, -1.954, 1.36] );
```

```
(%o1) C1 = 
$$\begin{bmatrix} 1.01 & 2.02 & 0.99 & 0.01 & 0.01 \\ 1.02 & 1.01 & -1.01 & 2.02 & 0.01 \\ 1.99 & 2.99 & 0.01 & 1.01 & 0.02 \\ -1.01 & 0.02 & 1.02 & -2.01 & 0.03 \\ 0.24 & -0.972 & 1.23 & -1.954 & 1.36 \end{bmatrix}$$

```

```
(%i2) detC1=determinant(C1);
```

```
(%o2) detC1 = 2.795694905840001
```

Com és conegut, la inversa d'una matriu quadrada es pot calcular aplicant el concepte de determinant, mitjançant la matriu d'adjunts o cofactors transposada multiplicada per l'invers del determinant de la matriu. Així, per exemple:

```
(%i1) 'M1=M1: matrix( [1,2,1,0], [1,1,-1,2], [2,3,0,1], [-1,0,1,-2] );
```

```
(%o1) M1 = 
$$\begin{bmatrix} 1 & 2 & 1 & 0 \\ 1 & 1 & -1 & 2 \\ 2 & 3 & 0 & 1 \\ -1 & 0 & 1 & -2 \end{bmatrix}$$

```

```
(%i2) detM1=detM1:determinant(M1);
```

```
(%o2) detM1 = 2
```

```
(%i3) adjM1=adjM1:adjoint(M1);
```

```
(%o3) adjM1 = 
$$\begin{bmatrix} -1 & -4 & 2 & -3 \\ 0 & 2 & 0 & 2 \\ 3 & 0 & -2 & -1 \\ 2 & 2 & -2 & 0 \end{bmatrix}$$

```

```
(%i4) inv_M1:=invM1:(1/detM1)*adjM1;
```

$$(\%o4) \text{ inv_M1} = \begin{bmatrix} -\frac{1}{2} & -2 & 1 & -\frac{3}{2} \\ 0 & 1 & 0 & 1 \\ \frac{3}{2} & 0 & -1 & -\frac{1}{2} \\ 1 & 1 & -1 & 0 \end{bmatrix}$$

Aquesta matriu és la inversa de la matriu M1; en efecte:

```
(%i5) M1.invM1;
```

$$(\%o5) \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

A-5.- Sistemes d'equacions lineals.

Els continguts d'aquest apartat es desenvolupen al fitxer **Apendix-5.wxm**.

A.5.1. Discussió i resolució de sistemes lineals.

La mecànica de resolució de sistemes d'equacions lineals senzills amb wxMaxima ja és coneguda. Recordem-la amb un exemple.

Exemple A.5.1. Resoldre el sistema d'equacions lineals:

$$\begin{cases} 2x - 4y = -2 \\ -3x + 5y = 1 \end{cases}$$

Introduïm les equacions i assignem a cada equació una referència:

```
(%i1) Eq1:2*x-4*y=-2;  
      Eq2:-3*x+5*y=1;  
(%o1) 2 x - 4 y = -2  
(%o2) 5 y - 3 x = 1
```

Per resoldre el sistema apliquem la instrucció “linsolve” indicant les equacions i les variables:

```
(%i3) linsolve([Eq1, Eq2], [x,y]);  
(%o3) [x = 3, y = 2]
```

Interessa ara il·lustrar el procediment de discussió i resolució de sistemes d'equacions lineals aplicant el teorema de Rouché-Frobenius. Ho farem amb dos exemples.

Exemple A.5.2. Discutir i resoldre el sistema d'equacions lineals:

$$\begin{cases} x + 2y + 3z = 1 \\ 2x - y + z = -1 \\ -x + 3y - 2z = 2 \end{cases}$$

Com s'ha dit abans, cal introduir les equacions i assignar a cada equació una referència:

```
(%i1) Eq1: x + 2*y + 3*z = 1$
      Eq2: 2*x - y + z = -1$
      Eq3: -x + 3*y - 2*z = 2$
```

Ara calculem amb wxMaxima les dues matrius associades al sistema lineal: la matriu del sistema formada pels coeficients de les incògnites o variables i a continuació la matriu ampliada, resultant de concatenar la matriu del sistema amb la columna dels termes independents. Matriu del sistema:

```
(%i4) A:=coefmatrix( [Eq1, Eq2, Eq3] , [x,y,z]);
```

$$(\%o4) A = \begin{bmatrix} 1 & 2 & 3 \\ 2 & -1 & 1 \\ -1 & 3 & -2 \end{bmatrix}$$

Matriu ampliada: hi ha una instrucció de wxMaxima que dóna aquesta matriu, però en una forma lleugerament diferent ja que ho fa passant al primer membre els termes independents i, per tant, canviant el seu signe. En efecte:

```
(%i5) A0:=augcoefmatrix( [Eq1, Eq2, Eq3] , [x,y,z]);
```

$$(\%o5) A0 = \begin{bmatrix} 1 & 2 & 3 & -1 \\ 2 & -1 & 1 & 1 \\ -1 & 3 & -2 & -2 \end{bmatrix}$$

Per calcular la matriu ampliada en la forma estàndard, cal definir el vector de termes independents i fer la concatenació amb la matriu del sistema. Per tant:

```
(%i6) load ("eigen")$ B:=B:columnvector([1, -1, 2])$
      A0:=A0:addcol( A, B) ;
```

$$(\%o8) A0 = \begin{bmatrix} 1 & 2 & 3 & 1 \\ 2 & -1 & 1 & -1 \\ -1 & 3 & -2 & 2 \end{bmatrix}$$

Ara s'ha de calcular el rang de cada una d'aquestes matrius:

```
(%i9) r_A:=rank(A); r_A0:=rank(A0);
(%o9) r_A = 3
(%o10) r_A0 = 3
```

Per tant el sistema és compatible determinat. Per resoldre'l per la metodologia clàssica es calcula la matriu esglaonada reduïda de la matriu ampliada:

```
(%i11) echelon(A0);
```

$$(\%o11) \begin{bmatrix} 1 & 2 & 3 & 1 \\ 0 & 1 & 1 & \frac{3}{5} \\ 0 & 0 & 1 & 0 \end{bmatrix}$$

D'aquí es pot obtenir la solució, així com amb la instrucció de wxMaxima a aquests efectes:

```
(%i12) linsolve( [Eq1, Eq2, Eq3] , [x,y,z] );
```

$$(\%o12) [x = -\frac{1}{5}, y = \frac{3}{5}, z = 0]$$

Exemple A.5.3. Discutir i resoldre el sistema d'equacions lineals:

$$\begin{cases} x + 2y + 3z = 1 \\ 2x - y + z = -1 \end{cases}$$

Introduïm les equacions i assignem a cada equació una referència:

```
(%i1) Eq1: x + 2*y + 3*z = 1$
      Eq2: 2*x - y + z = -1$
```

Ara calculem amb wxMaxima les dues matrius associades al sistema lineal. Matriu del sistema:

```
(%i3) A:=A:coefmatrix( [Eq1, Eq2] , [x,y,z]);
```

$$(\%o3) A = \begin{bmatrix} 1 & 2 & 3 \\ 2 & -1 & 1 \end{bmatrix}$$

Matriu ampliada:

```
(%i4) load ("eigen")$ B:=B:columnvector([1, -1])$
      A0:=A0:addcol( A, B) ;
```

$$(\%o6) A0 = \begin{bmatrix} 1 & 2 & 3 & 1 \\ 2 & -1 & 1 & -1 \end{bmatrix}$$

Ara calculem el rang de cada una d'aquestes matrius:

```
(%i7) r_A:=rank(A); r_A0:=rank(A0);
```

$$(\%o7) r_A = 2$$

$$(\%o8) r_{A0} = 2$$

Per tant el sistema és compatible i simplement indeterminat. Per resoldre'l apliquem la instrucció de wxMaxima:

```
(%i10) linsolve( [Eq1, Eq2] , [x,y,z] );
(%o10) [x = - $\frac{5\%r5+1}{5}$ , y = - $\frac{5\%r5-3}{5}$ , z = %r5]
```

És a dir la solució és:

$$\begin{cases} x = -\frac{5z+1}{5} \\ y = \frac{3-5z}{5} \\ z = z \in \mathbb{R} \quad (\text{parametre}) \end{cases}$$

Exemple A.5.4. Discutir i resoldre el sistema d'equacions lineals:

$$\begin{cases} x + 2y + 3z = 1 \\ 2x + 4y + 6z = -1 \end{cases}$$

Introduïm les equacions i assignem a cada equació una referència:

```
(%i1) Eq1: x + 2*y + 3*z = 1$
      Eq2: 2*x + 4*y + 6*z = -1$
```

Ara calculem amb wxMaxima les dues matrius associades al sistema lineal. Matriu del sistema:

```
(%i3) A:=A:coefmatrix( [Eq1, Eq2] , [x,y,z]);
(%o3) A =  $\begin{bmatrix} 1 & 2 & 3 \\ 2 & 4 & 6 \end{bmatrix}$ 
```

Matriu ampliada:

```
(%i4) load("eigen")$ B:=B:columnvector([1, -1])$
      A0:=A0:addcol( A, B );
(%o6) A0 =  $\begin{bmatrix} 1 & 2 & 3 & 1 \\ 2 & 4 & 6 & -1 \end{bmatrix}$ 
```

Ara calculem el rang de cada una d'aquestes matrius:

```
(%i7) r_A=rank(A);    r_A0=rank(A0);
(%o7) r_A = 1
(%o8) r_A0 = 2
```

Per tant el sistema és incompatible. Vegem què succeeix si es vol resoldre amb wxMaxima:

```
(%i10) linsolve( [Eq1, Eq2] , [x,y,z] );
(%o10) []
```

A.5.2. Sistemes lineals amb paràmetres.

La discussió d'un sistema d'equacions lineals sovint involucra l'existència d'un o més paràmetres en les equacions; la tipologia i la solució del sistema depèn del valor d'aquests paràmetres i cal fer la discussió corresponent. Il·lustrarem el procediment amb un parell d'exemples.

Exemple A.5.5. Discutir i resoldre en funció del paràmetre $a \in \mathbb{R}$ el sistema d'equacions lineals:

$$\begin{cases} 3x - 2y + z = 1 \\ 4x + y - 2z = 2 \\ 2x - 5y - az = 3 \end{cases}$$

Introduïm les equacions i assignem a cada equació una referència:

```
(%i1) Eq1: 3*x - 2*y + z = 1$
      Eq2: 4*x + y - 2*z = 2$
      Eq3: 2*x - 5*y - a*z = 3$
```

Matriu del sistema i matriu ampliada:

```
(%i4) A:=A:coefmatrix ( [Eq1, Eq2, Eq3] , [x,y,z] );
(%o4) A =  $\begin{bmatrix} 3 & -2 & 1 \\ 4 & 1 & -2 \\ 2 & -5 & -a \end{bmatrix}$ 

(%i5) load ("eigen")$    B=B:columnvector([1, 2, 3])$
      A0=A0:addcol( A, B) ;
(%o7) A0 =  $\begin{bmatrix} 3 & -2 & 1 & 1 \\ 4 & 1 & -2 & 2 \\ 2 & -5 & -a & 3 \end{bmatrix}$ 
```

Triangulitzem la matriu ampliada:

```
(%i8) triangularize(A0);
```

$$(\%o8) \begin{bmatrix} 3 & -2 & 1 & 1 \\ 0 & 11 & -10 & 2 \\ 0 & 0 & -11a-44 & 33 \end{bmatrix}$$

Ara tenim dues situacions:

- 1) Si $-11a - 44 = 0$, és a dir, si $a = -4$, aleshores $\text{rang}(A) = 2$; $\text{rang}(A_0) = 3$ i per tant el sistema és incompatible.
- 2) Si $-11a - 44 \neq 0$, és a dir, si $a \neq -4$, aleshores $\text{rang}(A) = \text{rang}(A_0) = 3$ i per tant el sistema és compatible determinat.

En efecte:

```
(%i9) A1=A1:matrix([3,-2,1],[4,1,-2],[2,-5,4]); r_A1=rank(A1);
```

$$(\%o9) A1 = \begin{bmatrix} 3 & -2 & 1 \\ 4 & 1 & -2 \\ 2 & -5 & 4 \end{bmatrix}$$

```
(%o10) r_A1=2
```

```
(%i11) A01=A01:matrix([3,-2,1,1],[4,1,-2, 2],[2,-5,4, 3]); r_A01=rank(A01);
```

$$(\%o11) A01 = \begin{bmatrix} 3 & -2 & 1 & 1 \\ 4 & 1 & -2 & 2 \\ 2 & -5 & 4 & 3 \end{bmatrix}$$

```
(%o12) r_A01=3
```

Solució del sistema en el segon cas ($a \neq -4$):

```
(%i13) linsolve([Eq1, Eq2, Eq3], [x,y,z]);
```

$$(\%o13) \left[x = \frac{5a+11}{11a+44}, y = \frac{2a-22}{11a+44}, z = -\frac{3}{a+4} \right]$$

Exemple A.5.6. Discutir i resoldre en funció dels paràmetres $a, b \in \mathbb{R}$ el sistema d'equacions lineals:

$$\begin{cases} ax + y + z = b \\ x + ay + z = b \\ x + y + az = b \end{cases}$$

Introduïm les equacions i assignem a cada equació una referència:

```
(%i1) Eq1: a*x + y + z = b$
      Eq2: x + a*y + z = b$
      Eq3: x + y + a*z = b$
```

Matriu del sistema i matriu ampliada:

```
(%i4) A=A:coefmatrix ( [Eq1, Eq2, Eq3] , [x,y,z] );
(%o4) A = 
$$\begin{bmatrix} a & 1 & 1 \\ 1 & a & 1 \\ 1 & 1 & a \end{bmatrix}$$

(%i5) load ("eigen")$ B=B:columnvector([b, b, b])$
      A0=A0:addcol( A, B ) ;
(%o7) A0 = 
$$\begin{bmatrix} a & 1 & 1 & b \\ 1 & a & 1 & b \\ 1 & 1 & a & b \end{bmatrix}$$

```

Discutirem el rang de la matriu del sistema ja que només depèn d'un paràmetre i ho farem amb el seu determinant, ja que és quadrada:

```
(%i8) detA=determinant(A);
      expand(%);
(%o8) detA = a (a2 - 1) - 2 a + 2
      (%o9) detA = a3 - 3 a + 2
(%i10) solve(a3 - 3*a + 2 = 0, a);
(%o10) [a = -2, a = 1]
```

Ara tenim tres situacions:

- 1) $a = 1$;
- 2) $a = -2$;
- 3) $a \neq 1$ i $a \neq -2$.

Primer cas. La matriu del sistema i la matriu ampliada són:

```
(%i11) A1=A1:matrix([1,1,1],[1,1,1],[1,1,1]);
(%o11) A1 = 
$$\begin{bmatrix} 1 & 1 & 1 \\ 1 & 1 & 1 \\ 1 & 1 & 1 \end{bmatrix}$$

```

```
(%i12) A01=A01:matrix([1,1,1,b],[1,1,1,b],[1,1,1,b]);
(%o12) A01 = 
$$\begin{bmatrix} 1 & 1 & 1 & b \\ 1 & 1 & 1 & b \\ 1 & 1 & 1 & b \end{bmatrix}$$

```

Rang d'aquestes matrius:

```
(%i13) r_A1=rank(A1);    r_A01=rank(A01);
(%o13) r_A1 = 1
(%o14) r_A01 = 1
```

Per tant el sistema és compatible doblement indeterminat. Solució del sistema:

```
(%i15) Eq11: x + y + z = b$
      Eq21: x + y + z = b$
      Eq31: x + y + z = b$
(%i18) linsolve ( [Eq11, Eq21, Eq31] , [x,y,z] );
solve: dependent equations eliminated: (2 3)
(%o18) [ x = b - %r7 - %r6, y = %r7, z = %r6 ]
```

Segon cas. La matriu del sistema i el seu rang són:

```
(%i19) A2=A2:matrix([-2,1,1],[1,-2,1],[1,1,-2]);    r_A2=rank(A2);
(%o19) A2 = 
$$\begin{bmatrix} -2 & 1 & 1 \\ 1 & -2 & 1 \\ 1 & 1 & -2 \end{bmatrix}$$

(%o20) r_A2 = 2
```

Matriu ampliada:

```
(%i21) A02=A02:matrix([-2,1,1,b],[1,-2,1,b],[1,1,-2,b]);
(%o21) A02 = 
$$\begin{bmatrix} -2 & 1 & 1 & b \\ 1 & -2 & 1 & b \\ 1 & 1 & -2 & b \end{bmatrix}$$

```

Triangulitzem la matriu:

```
(%i22) triangularize(A02);
```

```
(%o22) 
$$\begin{bmatrix} -2 & 1 & 1 & b \\ 0 & 3 & -3 & -3b \\ 0 & 0 & 0 & 9b \end{bmatrix}$$

```

Per tant si $b \neq 0$, el sistema és incompatible i si $b=0$, el sistema és compatible simplement indeterminat. Solució del sistema:

```
(%i23) Eq12: -2*x + y + z = 0$
```

```
Eq22: x -2*y + z = 0$
```

```
Eq32: x + y -2*z = 0$
```

```
(%i26) linsolve([Eq12, Eq22, Eq32], [x,y,z]);
```

```
solve: dependent equations eliminated: (3)
```

```
(%o26) [x = %r8, y = %r8, z = %r8]
```

Tercer cas. Se sap que el sistema és compatible determinat i aleshores es pot calcular ja la solució:

```
(%i27) linsolve([Eq1, Eq2, Eq3], [x,y,z] );
```

```
(%o27) 
$$\left[ x = \frac{b}{a+2}, y = \frac{b}{a+2}, z = \frac{b}{a+2} \right]$$

```